Comparing Short Volatility Strategies

Euan Sinclair, who has authored a couple of excellent books on volatility trading, also writes an interesting blog.
In one of his posts Euan conducts an experiment comparing the outcome of selling straddles vs strangles under a couple of different scenarios. In this post I am going to look at the analysis, to see if we can make a definitive determination as to which of the two option strategies is superior.

First Scenario - Volatility Declines

In the first of his scenarios Euan considers the case of a stock priced at $100 whose 1-year options are trading at an implied annual volatility of 40%. After selling either the ATM straddle or the 90/100 strangle, actual volatility in the stock over the next year falls to 30%. We would expect to make money using either option strategy, on average, and indeed this turns out to be the case.

Straddle

We first calculate the value of the ATM straddle, assuming no dividends in the stock, a zero interest rate and an implied volatility of 40%, using the standard Black-Scholes pricing model:

```
straddlePremium = 100 *
  (FinancialDerivative[{"European", "Call"}, {"StrikePrice" → 100, "Expiration" → 1},
     {"InterestRate" → 0.0, "CurrentPrice" → 100,
      "Dividend" → 0.0, "Volatility" → 0.4}, "Value"] +
    FinancialDerivative[{"European", "Put"}, {"StrikePrice" → 100,
      "Expiration" → 1}, {"InterestRate" → 0.0, "CurrentPrice" → 100,
      "Dividend" → 0.0, "Volatility" → 0.4}, "Value"])
```

```
3170.39
```

We'll assume Euan's estimates of the margin for the straddle and strangles:

```
straddleMargin = 2000; strangleMargin = 1000;
```

The maximum return on the straddle is therefore:

```
straddlePremium / straddleMargin
```

```
1.58519
```

If we forecast realized volatility over the ensuring year to be only 30%, our estimate of the fair value of the straddle will be:

```
straddleFairvalue = 100 *
   (FinancialDerivative[{"European", "Call"}, {"StrikePrice" → 100, "Expiration" → 1},
      {"InterestRate" → 0.0, "CurrentPrice" → 100,
       "Dividend" → 0.0, "Volatility" → 0.3}, "Value"] +
     FinancialDerivative[{"European", "Put"}, {"StrikePrice" → 100,
       "Expiration" → 1}, {"InterestRate" → 0.0, "CurrentPrice" → 100,
       "Dividend" → 0.0, "Volatility" → 0.3}, "Value"])
```

```
2384.71
```

Consequently, the expected return on the straddle sale is just over 39%:

```
(straddlePremium - straddleFairvalue) / straddleMargin
```

```
0.39284
```

We next simulate the path of the stock over monthly intervals a large number of times, assuming it follows a Geometric Brownian Motion process with zero drift and 30% annual volatility, and evaluate the outcome of the short straddle strategy in each path:

```
data =
 RandomFunction[GeometricBrownianMotionProcess[0, .3, 100], {0, 1, 1/12}, 10 000]
```

```
TemporalData[ Time: 0 to 1  Data points: 130 000  Paths: 10 000 ]
```

```
ListLinePlot[data]
```



We calculate the final stock price in each of the 10,000 paths:

```
simPrices = data[[2, 1]];
finalPrices = simPrices[[All, 13]];
```

We are now ready to evaluate the P&L from the short straddle strategy under each of the scenarios and look at the distribution of returns, as follows:

```
straddlePayoff = 100 * Abs[finalPrices - 100];
straddlePL = straddlePremium - straddlePayoff;
Histogram[StraddleReturns = straddlePL / straddleMargin,
 PlotLabel → Style["Straddle Return - Correct Forecast", Bold]]
```




```
straddleWinRate = Total[HeavisideTheta[straddlePL]] / Length[finalPrices] // N;
```

```
headings = {"Win Rate", "Mean", "Median",
   "Min", "Max", "St. Dev.", "Skewness", "Kurtosis"};
```

```
Distn =
  Flatten[{straddleWinRate, Through[{Mean, Median, Min, Max, StandardDeviation,
       Skewness, Kurtosis}[StraddleReturns]]}];
Grid[{headings, Distn}, Frame → All]
```

Win Rate	Mean	Median	Min	Max	St. Dev.	Skewness	Kurtosis
0.7291	0.393744	0.5818	-7.91848	1.58481	0.958705	-1.68262	8.14368

The average return across all 10,000 outcomes is 39%, exactly in line with our prediction, while almost 3/4 of trades are profitable. Under the worst case outcome the strategy loses over 790%, while the maximum gain exceeds 158%, again as predicted. The returns distribution is characterized by a very long left tail, featuring negative skewness and large kurtosis.

Strangle

Now let' s go through the same process as before, this time evaluating the outcome from selling a 90/110 strangle:

```
stranglePremium = 100 *
  (FinancialDerivative[{"European", "Call"}, {"StrikePrice" → 110, "Expiration" → 1},
     {"InterestRate" → 0.0, "CurrentPrice" → 100,
      "Dividend" → 0.0, "Volatility" → 0.4}, "Value"] +
    FinancialDerivative[{"European", "Put"}, {"StrikePrice" → 90, "Expiration" → 1},
     {"InterestRate" → 0.0, "CurrentPrice" → 100,
      "Dividend" → 0.0, "Volatility" → 0.4}, "Value"])
```

```
2267.94
```

The maximum return for the 90/110 strangle is:

```
stranglePremium / strangleMargin
```

```
2.26794
```

```
strangleFairValue = 100 *
  (FinancialDerivative[{"European", "Call"}, {"StrikePrice" → 110, "Expiration" → 1},
     {"InterestRate" → 0.0, "CurrentPrice" → 100,
      "Dividend" → 0.0, "Volatility" → 0.3}, "Value"] +
    FinancialDerivative[{"European", "Put"}, {"StrikePrice" → 90, "Expiration" → 1},
     {"InterestRate" → 0.0, "CurrentPrice" → 100,
      "Dividend" → 0.0, "Volatility" → 0.3}, "Value"])
```

```
1515.39
```

The expected return is higher for the strangle than for the straddle:

```
(stranglePremium - strangleFairValue) / strangleMargin
```

```
0.752552
```

The delta of the OTM 110 strike call is 0.48, while that of the 90 strike put is -0.32:

```
FinancialDerivative[{"European", "Call"},
 {"StrikePrice" → 110, "Expiration" → 1}, {"InterestRate" → 0.0,
  "CurrentPrice" → 100, "Dividend" → 0.0, "Volatility" → 0.4}, "Delta"]
```

```
0.484734
```

```
FinancialDerivative[{"European", "Put"},
 {"StrikePrice" → 90, "Expiration" → 1}, {"InterestRate" → 0.0,
  "CurrentPrice" → 100, "Dividend" → 0.0, "Volatility" → 0.4}, "Delta"]
```

```
-0.32154
```

Next we compute the payoff from the short strangle strategy and examine the distribution of returns:

```
callPayoff = (finalPrices - 110) * HeavisideTheta[finalPrices - 110];
putPayoff = (90 - finalPrices) * HeavisideTheta[90 - finalPrices];
```

```
stranglePayoff = 100 * (callPayoff + putPayoff);
stranglePL = stranglePremium - stranglePayoff;
Histogram[StrangleReturns = stranglePL / strangleMargin,
 PlotLabel → Style["Strangle Return - Correct Forecast", Bold]]
```




```
strangleWinRate = Total[HeavisideTheta[stranglePL]] / Length[finalPrices] // N;
Distn = Flatten[{strangleWinRate, Through[{Mean, Median, Min,
      Max, StandardDeviation, Skewness, Kurtosis}[StrangleReturns]]}];
Grid[{headings, Distn}, Frame → All]
```

Win Rate	Mean	Median	Min	Max	St. Dev.	Skewness	Kurtosis
0.7441	0.755725	1.26115	-15.7394	2.26794	1.79318	-2.04264	9.75708

Here too, for the strangle, the average and maximum returns are in line with their expected values. But notice that, although the average return is higher for the strangle than the straddle, the strangle suffers from double the size of maximum loss, while the distribution of returns has larger negative skewness and kurtosis. In other words, the tail risk is much greater for the short strangle than for the short straddle.

Second Scenario - Volatility Rises

In Euan' s second scenario the volatility forecast turns out to be wrong and rather than falling from 40% to 30%, volatility instead rises to 70%. We further suppose an upward drift in the stock of 20% per annum.

As before, we generate a large number of sample paths:

```
data =
 RandomFunction[GeometricBrownianMotionProcess[0.2, 0.7, 100], {0, 1, 1/12}, 10 000]
```

```
TemporalData[ Time: 0 to 1  Data points: 130000  Paths: 10000 ]
```

```
ListLinePlot[data]
```



Notice that the stock exhibits much greater volatility and attains much higher price levels than in the original scenario due to the upward trend in the stock. We calculate the final prices after one year, as before:

```
simPrices = data[[2, 1]];
finalPrices = simPrices[[All, 13]];
```

Straddle

```
straddlePayoff = 100 * Abs[finalPrices - 100];
straddlePL = straddlePremium - straddlePayoff;
Histogram[StraddleReturns = straddlePL / straddleMargin,
 PlotLabel → Style["Straddle Return - Incorrect Forecast", Bold]]
```



```
straddleWinRate = Total[HeavisideTheta[straddlePL]] / Length[finalPrices] // N;
Distn =
  Flatten[{straddleWinRate, Through[{Mean, Median, Min, Max, StandardDeviation,
       Skewness, Kurtosis}[StraddleReturns]]}];
Grid[{headings, Distn}, Frame → All]
```

Win Rate	Mean	Median	Min	Max	St. Dev.	Skewness	Kurtosis
0.3688	-1.53106	-0.60917	-51.2786	1.58433	3.71575	-3.85416	26.3208

Under the second scenario the short straddle loses money, on average, while only around 1/3 of all trades are profitable. This is to be expected, given that our volatility forecast was so poor. Furthermore, the left-tail risk of the strategy has clearly increased, while the returns distribution exhibits a larger negative skewness and kurtosis than before.

Strangle

```
callPayoff = (finalPrices - 110) * HeavisideTheta[finalPrices - 110];
putPayoff = (90 - finalPrices) * HeavisideTheta[90 - finalPrices];
```

```
stranglePayoff = 100 * (callPayoff + putPayoff);
stranglePL = stranglePremium - stranglePayoff;
Histogram[StrangleReturns = stranglePL / strangleMargin,
 PlotLabel → Style["Strangle Return - Incorrect Forecast", Bold]]
```




```
strangleWinRate = Total[HeavisideTheta[stranglePL]] / Length[finalPrices] // N;
Distn = Flatten[{strangleWinRate, Through[{Mean, Median, Min,
      Max, StandardDeviation, Skewness, Kurtosis}[StrangleReturns]]}];
Grid[{headings, Distn}, Frame → All]
```

Win Rate	Mean	Median	Min	Max	St. Dev.	Skewness	Kurtosis
0.3789	-3.02306	-1.12079	-102.46	2.26794	7.38731	-3.91351	26.8183

While the skewness and kurtosis of the returns distribution for the strangle and straddle are approximately the same under this scenario, the average PL, median PL and maximum loss are all much worse for the strangle, compared to the straddle.

Conclusion

We conclude, as did Euan in his original analysis, that the straddle is superior to the strangle as a strategy for selling volatility. The investor might be encouraged to believe that the strangle is less risky, because the initial price of the stock is some distance from the option strike prices. However, it turns out that the higher average returns of the strangle under benign market conditions comes at the cost of greater downside risk.   Under adverse market conditions, the performance of the straddle is much superior to that of the strangle, which produces average and median losses that are almost double that of the straddle.